

ATENÇÃO! As atividades práticas serão por meio de atividades já solucionadas. Nossa ideia é que você entenda o contexto e recrie o código-fonte apresentado em seu Portugol Studio. Com isso, você terá exemplos mais robustos para poder seguir sua evolução.

ATIVIDADE 1

Construiremos um programa que perguntará ao usuário a maioridade civil do país que ele mora. Depois, ele vai informar sua idade. Nossa missão será dizer apenas se ele é ou não maior de idade.

Figura 1 - Atividade 1

```
programa
{
    funcao inicio()
    {
        inteiro maioridadeCivil
        inteiro idade

        escreva("Olá!\n")
        escreva("Nesse programa verificaremos se você é maior de idade ou não.\n")

        escreva("Informe a maioridade civil do país que você vive: ")
        leia(maioridadeCivil)
        escreva("\n")

        se(maioridadeCivil < 0)
        {
            escreva("A maioridade civil informada é negativa. Por isso, o programa será encerrado.")
        } senao {

            escreva("Informe a sua idade: ")
            leia(idade)
            escreva("\n")

            se(idade < 0)
            {
                escreva("A idade informada é negativa. Por isso, o programa será encerrado.")
            } senao {

                se(idade < maioridadeCivil)
                {
                    escreva("Você é menor de idade.")
                } senao {
                    escreva("Você é maior de idade.")
                }
            }
        }

        escreva("\n")
    }
}
```

Fonte: Autores, 2023.

ATIVIDADE 2

Agora, faremos uma solução para um mercado. Eles vendem maçãs e querem aplicar um desconto para o cliente. O preço normal das maçãs é R\$ 1.30. Porém, a partir de uma dúzia, ou 12 maçãs, é pago apenas R\$ 1.00 por unidade. A ideia nossa é mostrar quantas maçãs foram compradas e qual o valor que será pago ao final.

Figura 2 - Atividade 2

```
programa
{
    funcao inicio()
    {
        inteiro qtdMacas
        real valorMacas
        real valorFinal

        escreva("Olá!\n")
        escreva("Nesse programa faremos o cálculo do valor de venda de maçãs.\n")

        escreva("Informe a quantidade de maçãs vendidas: ")
        leia(qtdMacas)
        escreva("\n")

        se(qtdMacas <= 0)
        {
            escreva("A quantidade de maçãs vendidas é zero ou negativa. Por isso, o programa será encerrado.")
        } senao {

            se(qtdMacas < 12)
            {
                valorMacas = 1.30
            } senao {
                valorMacas = 1.00
            }

            valorFinal = qtdMacas * valorMacas

            escreva("O valor final de venda das maçãs é de R$ " + valorFinal)

        }

        escreva("\n")
    }
}
```

Fonte: Autores, 2023.

ATIVIDADE 3

Pensaremos numa solução que pedirá ao usuário um número inteiro, o seu antecessor (aquele que vem antes) e seu sucessor (aquele que vem depois). Depois, verificaremos se os dados informados estão corretos, validando essa informação a partir do número informado. Se ele acertou, venceu o desafio. Se não, saiu derrotado.

EXEMPLO

Número informado = 3
Número antecessor informado = 2
Número sucessor informado = 4

Número antecessor real = 2
Número sucessor real = 3

O usuário perdeu o desafio

Figura 3 - Atividade 3

```
programa
{
    funcao inicio()
    {
        inteiro nroInformado
        inteiro nroAntecessor
        inteiro nroSucessor

        escreva("Olá!\n")
        escreva("Nesse programa verificaremos se o número antecessor e o sucessor estão corretos.\n")

        escreva("Informe o número a ser verificado: ")
        leia(nroInformado)
        escreva("\n")

        escreva("Informe seu número antecessor: ")
        leia(nroAntecessor)

        escreva("Informe seu número sucessor: ")
        leia(nroSucessor)

        escreva("\n")

        se((nroAntecessor == (nroInformado - 1)) e (nroSucessor == (nroInformado + 1)))
        {
            escreva("VOCÊ VENCEU!!!")
        } senao {
            escreva("Infelizmente, você perdeu.")
        }

        escreva("\n")
    }
}
```

Fonte: Autores, 2023.