

ATENÇÃO! As atividades práticas serão por meio de atividades já solucionadas. Nossa ideia é que você entenda o contexto e recrie o código-fonte apresentado em seu Portugol Studio. Com isso, você terá exemplos mais robustos para poder seguir sua evolução.

ATIVIDADE 1

Vamos construir um algoritmo simples para começarmos a conhecer melhor o funcionamento de uma matriz. Lembre-se que a matriz pode ser entendida como uma estrutura com linhas e colunas. Ou ainda, é um vetor de vetor. Portanto, criaremos uma matriz simples com 2 (duas) linhas e 2 colunas (duas), ou simplesmente 2x2.

Figura 1 - Atividade 1

```
programa
{
    funcao inicio()
    {
        inteiro matriz[2][2]

        para(inteiro i=0;i<2;i++){
            para(inteiro j=0;j<2;j++){
                escreva((i+j+1)+"º Valor: ")
                leia(matriz[i][j])
            }
        }
        limpa()
        para(inteiro i=0;i<2;i++){
            para(inteiro j=0;j<2;j++){
                escreva("\n"+(i+j+1)+"º Valor: "+matriz[i][j])
            }
        }
    }
}
```

Fonte: Autores, 2023.

ATIVIDADE 2

Agora, criaremos um programa formado por uma matriz 3x3. Alimentaremos essa matriz com valores inteiros informados pelo usuário. Ao final, solicitaremos ao próprio usuário uma das linhas e mostraremos os números contidos nela. É importante salientar que só aceitaremos que sejam informadas as linhas 1, 2 ou 3. Qualquer valor além ou aquém fará com que a solicitação seja feita novamente.

Figura 2 - Atividade 2

```
programa
{
    funcao inicio()
    {
        inteiro matriz[3][3], linha

        para(inteiro i=0;i<3;i++){
            para(inteiro j=0;j<3;j++){
                escreva((i+j+1)+"º Valor: ")
                leia(matriz[i][j])
            }
        }
        faca{
            escreva("\n\nLinha da matriz desejada (1 a 3): ")
            leia(linha)
        }enquanto(linha<1 ou linha>3)

        escreva(matriz[linha][0]+" | "+matriz[linha][1]+" | "+matriz[linha][2])

    }
}
```

Fonte: Autores, 2023.

ATIVIDADE 3

Nessa atividade, criaremos uma matriz retangular, com as dimensões 2x3. Nela, o usuário informará todos os valores inteiros que desejará para compor essa matriz. Ao final, será feita a soma de todos os elementos da matriz e mostrado ao usuário.

Figura 3 - Atividade 3

```
programa
{
    funcao inicio()
    {
        inteiro matriz[2][3], soma=0, cont=1

        para(inteiro i=0;i<2;i++){
            para(inteiro j=0;j<3;j++){
                escreva(cont+"º Valor: ")
                leia(matriz[i][j])
                soma += matriz[i][j]
                cont++
            }
        }
        escreva("Soma: "+soma)
    }
}
```

Fonte: Autores, 2023.

ATIVIDADE 4

Aqui será criado um programa que carregará os valores de vendas de uma loja ao longo do primeiro semestre de 2022. Para isso, será construída uma matriz 6x4. Com isso, serão representados os primeiros 6 meses do ano, sendo que cada mês é formado por 4 semanas (6 meses x 4 semanas). Ao final, será mostrado o faturamento da loja da seguinte forma:

- Total de vendas do semestre
- Total vendido em cada mês

Figura 4 - Atividade 4

```
programa
{
    funcao inicio()
    {
        real vendas[6][4], totalVendasMes[6], totalVendasSemestre=0

        para(inteiro i=0;i<6;i++){
            para(inteiro j=0;j<4;j++){
                escreva("Venda da " + (j+1) + "ª semana do " + (i+1) + "º mês: ")
                leia(vendas[i][j])

                totalVendasMes[i] += vendas[i][j]
            }
            totalVendasSemestre += totalVendasMes[i]
        }

        limpa()

        para(inteiro i=0;i<6;i++){
            escreva("\nVendas do " + (i+1) + "º mês: " + totalVendasMes[i])
        }
        escreva("\n\nVendas totais do semestre: " + totalVendasSemestre)

    }
}
```

Fonte: Autores, 2023.