

*ATENÇÃO! As atividades práticas serão por meio de atividades já solucionadas. Nossa ideia é que você entenda o contexto e recrie o código-fonte apresentado em seu Portugol Studio. Com isso, você terá exemplos mais robustos para poder seguir sua evolução.*

## ATIVIDADE 1

Construiremos um programa inicial mais simples que solicitará apenas o ano de nascimento de 5 (cinco) pessoas e mostrará quais delas são a mais velha e a mais nova.

Figura 1- Atividade 1

```
programa
{
    funcao inicio()
    {
        inteiro nroIdade[5];
        inteiro maisNovo, maisVelho;

        maisNovo = 0
        maisVelho = 0

        para(inteiro i = 0; i < 5; i++)
        {
            escreva("Insira a idade da " + (i+1) + "ª pessoa: ");
            leia(nroIdade[i]);

            se (i == 0) { // Tente entender o por que de fazermos isso e discuta com os professor a respeito
                maisVelho = nroIdade[i]
                maisNovo = nroIdade[i]
            }

            se (maisVelho < nroIdade[i]) {
                maisVelho = nroIdade[i]
            }

            se (maisNovo > nroIdade[i]) {
                maisNovo = nroIdade[i]
            }
        }

        msgFinal(maisVelho, maisNovo)
    }

    funcao msgFinal(inteiro maisVelho, inteiro maisNovo)
    {
        escreva("\nA pessoa mais velha tem " + maisVelho + " anos")
        escreva("\nA pessoa mais nova tem " + maisNovo + " anos")
    }
}
```

Fonte: Autores, 2023.

## ATIVIDADE 2

Construiremos um programa que receberá 3 (três) notas de um aluno e faremos o cálculo da média aritmética dessas notas. Caso o aluno tire uma nota maior do que 7 (sete) ele está aprovado, porém, se a nota for abaixo ele estará reprovado. Um ponto importante é que a nota deverá ser arredondada em 2 (duas) casas. Ou seja, 3,3333333 será convertido para 3,33, por exemplo.

Figura 2 - Atividade 2

```
programa
{
    inclui biblioteca Matematica

    funcao inicio()
    {
        real notas[3]

        escreva("Esse programa tem a finalidade de calcular a média das notas do aluno.")

        para (inteiro i=0; i<3; i++)
        {
            notas[i] = insereNota(i)
        }

        calculaMedia(notas)
    }

    funcao real insereNota(inteiro ordem)
    {
        real notaAvaliacao

        escreva("Informe a " + (ordem+1) + " nota: ")
        leia(notaAvaliacao)

        retorne notaAvaliacao
    }

    funcao calculaMedia(real notas[])
    {
        real somaNotas, mediaNotas

        somaNotas = 0.0

        para (inteiro i=0; i<3; i++)
        {
            somaNotas += notas[i]
        }

        mediaNotas = (somaNotas / 3)
        mediaNotas = Matematica.arredondar(mediaNotas, 2)

        se (mediaNotas >= 7)
        {
            escreva("\nEsse aluno foi aprovado com média " + mediaNotas + "!")
        } senao {
            escreva("\nInfelizmente esse aluno foi reprovado com média " + mediaNotas + ".")
        }
    }
}
```

Fonte: Autores, 2023.

### ATIVIDADE 3

Agora complementaremos a atividade acima, deixando-a mais complexa. Perguntaremos se o professor quer uma média aritmética ou ponderada.

A nota aritmética é aquela clássica: se tenho 3 (três) notas, divido a soma das minhas notas por 3 (três). Por exemplo:

MÉDIA ARITMÉTICA	
Nota 1 = 7.1	Nota 2 = 8.3    Nota 3 = 6.6
Soma das notas = 22	Quantidade de notas = 3
Média = Soma das notas / Quantidade de notas	
Média = $22 / 3 = 7.333333333333333$	
Média arredondada = 7.33	

A nota ponderada exige que se saiba o peso de cada nota, ou seja, a importância de cada avaliação. Por exemplo:

MÉDIA PONDERADA	
Nota 1 = 7.1	Nota 2 = 8.3    Nota 3 = 6.6
Peso da nota 1 = 1	Peso da nota 2 = 2    Peso da nota 3 = 3
Cálculo da nota 1 = $7.1 * 1 = 7.1$	
Cálculo da nota 2 = $8.3 * 2 = 16.6$	
Cálculo da nota 3 = $6.6 * 3 = 19.8$	
Soma dos cálculos das notas = 43,5	
Soma dos pesos = 6	
Média = Soma dos cálculos das notas / Soma dos pesos	
Média = $43.5 / 6 = 7.25$	
Média arredondada = 7.25	

Agora sim! Depois de toda essa explicação, partiremos para a solução do nosso problema!

Figura 3 - Atividade (parte 1)

```
programa
{
    inclui biblioteca Matematica

    funcao inicio()
    {
        real notas[3]
        real mediaNotas = 0.0
        inteiro tipoMedia
        real peso[3]

        escreva("Esse programa tem a finalidade de calcular a média das notas do aluno.\n")

        para (inteiro i=0; i<3; i++)
        {
            notas[i] = insereNota(i)
        }

        escreva("\nQual o tipo de média você gostaria de calcular? Digite 1 para o cálculo aritmético; ou 2 para o cálculo ponderado: ")
        leia(tipoMedia)

        escolha (tipoMedia)
        {
            caso 1:
                escreva("\nVocê selecionou o cálculo da média aritmética.")
                mediaNotas = mediaAritmetica(notas)
                pare

            caso 2:
                escreva("\nVocê selecionou o cálculo da média ponderada.")

                para (inteiro i=0; i<3; i++)
                {
                    escreva("\nInforme o peso da " + (i+1) + "ª nota: ")
                    leia(peso[i])
                }

                mediaNotas = mediaPonderada(notas, peso)
                pare

                caso contrario:
                    escreva("\nSelecione uma opção válida.")
                    pare
        }

        mediaNotas = Matematica.arredondar(mediaNotas, 2)

        avaliacaoMedia(mediaNotas)
    }
}
```

Fonte: Autores, 2023.

Figura 4 - Atividade 3 (parte 2)

```
funcao real insereNota(inteiro ordem)
{
    real notaAvaliacao

    escreva("Informe a " + (ordem+1) + "ª nota: ")
    leia(notaAvaliacao)

    retorne notaAvaliacao
}

funcao real mediaAritmetica(real notas[])
{
    real somaNotas, mediaNotas

    somaNotas = 0.0

    para (inteiro i=0; i<3; i++)
    {
        somaNotas += notas[i]
    }

    mediaNotas = (somaNotas / 3)

    retorne mediaNotas
}

funcao real mediaPonderada(real notas[], real peso[])
{
    real mediaNotas
    mediaNotas = (notas[0] * peso[0] + notas[1] * peso[1] + notas[2] * peso[2]) / (peso[0] + peso[1] + peso[2])

    retorne mediaNotas
}
```

Fonte: Autores, 2023.

Figura 5 - Atividade 3 (parte 3)

```
funcao avaliacaoMedia(real mediaFinal)
{
    se (mediaFinal >= 7)
    {
        escreva("\nEsse aluno foi aprovado com média " + mediaFinal + "!")
    } senao {
        escreva("\nInfelizmente esse aluno foi reprovado com média " + mediaFinal + ".")
    }
}
}
```

Fonte: Autores, 2023.